Лабораторная работа №3

Метод неопределенных коэффициентов для нахождения частного и остатка от деления многочлена на многочлен

БГУ, ММФ, 1 курс
доцент Щеглова Н.Л.
сентябрь 2021 г.

Задание 1. Частное и остаток от деления многочлена на многочлен

Написать пользовательскую функцию, вычисляющую частное и остаток от деления многочлена $f_n(x)$ на многочлен $g_m(x)$, где n и m -- степени многочленов, $n \geqslant m$.
Использовать метод неопределенных коэффициентов.

Ключевые сведения

Выполнение задания 1

Задание 1.1 Функция, создающая многочлен с неопределенными коэффициентами

Построить функцию **Poly**, которая по заданной степени, целому неотрицательному числу, строит многочлен от одной переменной с неопределенными коэффициентами. Проектируемая функция **Poly** должна также предоставить возможность указания имени переменной и имени неопределенных коэффициентов.

Выполнение Задания 1.1.

Построим многочлен с неопределенными коэффициентами, используя функции **Map** и **Apply**.

Мономы определим, используя безымянную, или чистую функцию ***a*[#] *x*^# &**.

```
Plus @@ (a[#] x^# & /@ Range[0, 7])
```
сложить　　　　диапазон

$a[0] + x\, a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4] + x^5 a[5] + x^6 a[6] + x^7 a[7]$

Рассмотрим подробно, по шагам, как *Mathematica* вычисляет представленную выше суперпозицию функций. Вычисления проводим, рассматривая выражение из ячейки In[1] как сложную функцию.

```
Range[0, 7]
```
диапазон

{0, 1, 2, 3, 4, 5, 6, 7}

```
a[#] x^# & /@ {0, 1, 2, 3, 4, 5, 6, 7}
```

$\{a[0], x\, a[1], x^2 a[2], x^3 a[3], x^4 a[4], x^5 a[5], x^6 a[6], x^7 a[7]\}$

```
a[#] x^# & /@ Range[0, 7]
```
диапазон

{a[0], x a[1], x^2 a[2], x^3 a[3], x^4 a[4], x^5 a[5], x^6 a[6], x^7 a[7]}

```
Plus @@ {a[0], x a[1], x^2 a[2], x^3 a[3], x^4 a[4], x^5 a[5], x^6 a[6], x^7 a[7]}
```
сложить

a[0] + x a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4] + x^5 a[5] + x^6 a[6] + x^7 a[7]

Теперь позаботимся об аргументах функции ***Poly***, предусмотрев возможность задать следующие значения по умолчанию:
имя переменной ***x***,
имена неопределенных коэффициентов ***a[i], i=1, ..., n***.

```
Poly[n_Integer?NonNegative, x_Symbol : x, a_Symbol : a] := Plus @@ (a[#] x^# & /@ Range[0, n])
```
введённые··· неотрицательный сложить диапазон

Проверим введенное посредством ***Set*** (**:=**) глобальное правило, его *Mathematica* ассоциировала с символом ***Poly***

```
? Poly
```

Global`Poly

```
Poly[n_Integer?NonNegative, x_Symbol : x, a_Symbol : a] := Plus @@ (a[#1] x^#1 &) /@ Range[0, n]
```

Тестируем работу функции ***Poly***

```
Poly[7]
```

a[0] + x a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4] + x^5 a[5] + x^6 a[6] + x^7 a[7]

```
Poly[7, t]
```

a[0] + t a[1] + t^2 a[2] + t^3 a[3] + t^4 a[4] + t^5 a[5] + t^6 a[6] + t^7 a[7]

```
Poly[7, x, b]
```

b[0] + x b[1] + x^2 b[2] + x^3 b[3] + x^4 b[4] + x^5 b[5] + x^6 b[6] + x^7 b[7]

```
Poly[0]
```

a[0]

Задание 1.2 Частное и остаток от деления как многочлены с неопределенными коэффициентами

Записать частное и остаток от деления многочлена $f_n(x)$ на многочлен $g_m(x)$ в виде многочленов **quo** и **rem** с неопределенными коэффициентами.

Выполнение Задания 1.2.

Выполним задание 1.2 на конкретном примере, по шагам, всякий раз анализируя результат вычислений.

Пусть заданы: $f_n(x)$ – делимое, многочлен степени n,
$g_m(x)$ – делитель, многочлен степени m, $n \geqslant m$.

```
f = x^5 - 6 x^3 + 2 x^2 - 4;
g = x^2 - x + 1;
```

Для вычисления частного от деления многочленов $f_n(x)$ на $g_m(x)$ построим многочлен с неопределенными коэффициентами ***quo***.
Используем функцию ***Poly***, указав в качестве аргумента степень многочлена ***quo***, она равна n - m.

```
Exponent[f, x]
```
степень многочлена

5

```
Exponent[g, x]
```
степень многочлена

2

```
quo = Poly[Exponent[f, x] - Exponent[g, x]]
```
степень многочлена степень многочлена

$a[0] + x\,a[1] + x^2\,a[2] + x^3\,a[3]$

Аналогичным образом построим многочлен ***rem***,
который будет представлять остаток от деления многочленов $f_n(x)$ на $g_m(x)$, его степень не превосходит *m-1*.

```
rem = Poly[Exponent[g, x] - 1, x, b]
```
степень многочлена

$b[0] + x\,b[1]$

Теперь задача вычисления частного и остатка от деления многочлена $f_n(x)$ на многочлен $g_m(x)$ свелась к отысканию неопределенных коэффициентов многочленов ***quo*** и ***rem***.

Задание 1.3 Вычисление неопределенных коэффициентов многочленов

Для заданных $f_n(x)$ и $g_m(x)$ вычислить неопределенные коэффициенты многочленов **quo** и **rem** такими, чтобы условие
$f_n(x) - (g_m(x) quo + rem) = 0$ выполнялось для любых значений переменной x.

Выполнение Задания 1.3.

Согласно условию задачи, многочлен ***f - (g*quo + rem)*** должен тождественно равняться нулю.

```
f - (g quo + rem)
```

$-4 + 2x^2 - 6x^3 + x^5 - (1 - x + x^2)(a[0] + x\,a[1] + x^2\,a[2] + x^3\,a[3]) - b[0] - x\,b[1]$

```
f - (g quo + rem) // Expand
```
раскрыть скобки

$-4 + 2x^2 - 6x^3 + x^5 - a[0] + x\,a[0] - x^2\,a[0] - x\,a[1] + x^2\,a[1] - x^3\,a[1] - x^2\,a[2] + x^3\,a[2] - x^4\,a[2] - x^3\,a[3] + x^4\,a[3] - x^5\,a[3] - b[0] - x\,b[1]$

```
f - (g quo + rem) // Collect[#, x] &
```
сгруппировать

$-4 - a[0] + x^2\,(2 - a[0] + a[1] - a[2]) + x^5\,(1 - a[3]) + x^3\,(-6 - a[1] + a[2] - a[3]) + x^4\,(-a[2] + a[3]) - b[0] + x\,(a[0] - a[1] - b[1])$

Условие тождественного равенства нулю многочлена ***f - (g*quo + rem)***
приводит к решению системы линейных алгебраических уравнений (СЛАУ) относительно искомых неопределенных коэффициентов.
СЛАУ составляется следующим образом: группируются коэффициенты многочлена при одинаковых степенях переменной и для каждого коэффициента (множителя при x^i, $i=0,...,n$) записывается условие равенства нулю этого коэффициента. Чтобы многочлен был тождественно равным нулю, должно выполняться каждое записанное условие - получаем систему уравнений.

Составим такую систему

```
temp1 = CoefficientList[f - (g quo + rem), x]
```
список коэффициентов многочлена

```
{-4 - a[0] - b[0], a[0] - a[1] - b[1],
 2 - a[0] + a[1] - a[2], -6 - a[1] + a[2] - a[3], -a[2] + a[3], 1 - a[3]}
```

```
temp2 = Select[temp1, # =!= 0 &]
        выбрать
```

```
{-4 - a[0] - b[0], a[0] - a[1] - b[1],
 2 - a[0] + a[1] - a[2], -6 - a[1] + a[2] - a[3], -a[2] + a[3], 1 - a[3]}
```

```
temp3 = (# == 0 & /@ temp2)
```

```
{-4 - a[0] - b[0] == 0, a[0] - a[1] - b[1] == 0, 2 - a[0] + a[1] - a[2] == 0,
 -6 - a[1] + a[2] - a[3] == 0, -a[2] + a[3] == 0, 1 - a[3] == 0}
```

и решим ее, предварительно подготовив список ***vars***, содержащий неопределенные коэффициенты

```
vars = CoefficientList[#, x] & /@ {quo, rem} // Flatten
       список коэффициентов многочлена         уплостить
```

```
{a[0], a[1], a[2], a[3], b[0], b[1]}
```

```
temp4 = Solve[temp3, vars]
        решить уравнения
```

```
{{a[0] → -5, a[1] → -6, a[2] → 1, a[3] → 1, b[0] → 1, b[1] → 1}}
```

Остается подставить вычисленные значения коэффициентов в искомые многочлены

```
{quo, rem} /. temp4
```

$\{\{-5 - 6x + x^2 + x^3, 1 + x\}\}$

Задание 1.4 Вычисление неопределенных коэффициентов многочленов с помощью суперпозиции безымянных функций

Написать чистую функцию-однострочник, которая для заданных многочленов $f_n(x)$ и $g_m(x)$ вычисляет неопределенные коэффициенты многочленов **quo** и **rem** такими, чтобы условие $f_n(x)$ - $(g_m(x)quo + rem) = 0$ выполнялось для любых значений переменной x.

Выполнение Задания 1.4.

Оформим выполненную ранее последовательность действий в виде одного выражения, такое выражение называют функцией-однострочником.

```
f - (g quo + rem) // CoefficientList[#, x] & //
                    список коэффициентов многочлена
    Select[#, # =!= 0 &] & //
    выбрать
   (# == 0 & /@ #) & //
  Solve[#, vars] & //
  решить уравнения
 ({quo, rem} /. #) & // First
                        первый
```

$\{-5 - 6x + x^2 + x^3, 1 + x\}$

Следует отметить, что при работе с многочленом, коэффициенты которого содержат параметры, задача выбора тех значений параметров, при которых этот многочлен тождественно равен нулю, встречается довольно часто.

Сформулируем эту задачу отдельно и решим ее, построив функцию пользователя.

Задание 1.5 Пользовательская функция вычисления частного и остатка при делении многочлена на многочлен

Пусть задан многочлен *poly* одной переменной, коэффициентами *poly* являются выражения, содержащие параметры.
Постройте функцию **QuoRem**, которая вычисляет те значения параметров, при которых многочлен *poly* тождественно равен нулю.

Выполнение Задания 1.5.

Построим вспомогательную функцию **PolyCoeffSolve[poly, x, vars]** , которая для заданного многочлена **poly** от переменной **x** вычисляет те значения параметров **vars**, содержащихся в коэффициентах многочлена, при которых **poly** тождественно равен нулю.

```
PolyCoeffSolve[poly_, x_Symbol, vars_List] := poly //
      CoefficientList[#, x] & //
      список коэффициентов многочлена
     Select[#, # =!= 0 &] & //
     выбрать
    (# == 0 & /@ #) & //
   Solve[#, vars] & // First
   решить уравнения    первый
```

```
? PolyCoeffSolve
```

Global`PolyCoeffSolve

```
PolyCoeffSolve[poly_, x_Symbol, vars_List] := First[(Solve[#1, vars] &)[
   ((#1 == 0 &) /@ #1 &)[(Select[#1, #1 =!= 0 &] &)[(CoefficientList[#1, x] &)[poly]]]]]
```

```
PolyCoeffSolve[f - (g quo + rem), x, vars]
```

```
{a[0] → -5, a[1] → -6, a[2] → 1, a[3] → 1, b[0] → 1, b[1] → 1}
```

Используем созданную функцию **PolyCoeffSolve** для построения искомой функции **QuoRem**

```
quo = Poly[Exponent[f, x] - Exponent[g, x]]
          степень многочлена степень многочлена
```

$a[0] + x\, a[1] + x^2 a[2] + x^3 a[3]$

```
rem = Poly[Exponent[g, x] - 1, x, b]
          степень многочлена
```

$b[0] + x\, b[1]$

```
vars = CoefficientList[#, x] & /@ {quo, rem} // Flatten
       список коэффициентов многочлена             уплостить
```

```
{a[0], a[1], a[2], a[3], b[0], b[1]}
```

```
QuoRem[f_, g_, x_Symbol] :=
 Block[{quo, rem, vars}, quo = Poly[Exponent[f, x] - Exponent[g, x]];
 программный блок                   степень многочлена степень многочлена
  rem = Poly[Exponent[g, x] - 1, x, b];
            степень многочлена
  vars = CoefficientList[#, x] & /@ {quo, rem} // Flatten;
         список коэффициентов многочлена             уплостить
  PolyCoeffSolve[f - (g quo + rem), x, vars] // ({quo, rem} /. #) &]
```

```
? QuoRem
```

Global`QuoRem

```
QuoRem[f_, g_, x_Symbol] := Block[{quo, rem, vars}, quo = Poly[Exponent[f, x] - Exponent[g, x]];
  rem = Poly[Exponent[g, x] - 1, x, b];
  vars = Flatten[(CoefficientList[#1, x] &) /@ {quo, rem}];
  ({quo, rem} /. #1 &)[PolyCoeffSolve[f - (g quo + rem), x, vars]]]
```

Убедимся в правильности результата

```
result = QuoRem[f, g, x]
```

$\{-5 - 6x + x^2 + x^3, 1 + x\}$

```
f - (g result[[1]] + result[[2]]) // Simplify
```

упростить

```
0
```

Задание 2. Встроенные функции полиномиальной алгебры

Задание 2.1 Вычислите частное и остаток от деления многочлена *f* на многочлен *g*

Выполните предложенный вариант задания, используя встроенные функции *Mathematica* **PolynomialQuotient**, **PolynomialRemainder**, **PolynomialQuotientRemainder**. Результат вычислений оформите в виде таблицы.

1 : "f:" $3 - 2c + 2c^2 + 2c^3 - 5c^4 - c^5 + c^6 - 4c^7$
"g:" $3 - 5c - 5c^2 - 4c^3$

2 : "f:" $-3 + 5b - 3b^2 + b^3 - 5b^4 + 3b^5 + 2b^6 + 4b^7$
"g:" $2b - b^2 + 2b^3$

3 : "f:" $-3 - 4z + 2z^3 - 5z^4 + 4z^5 - 4z^6 - 5z^7$
"g:" $-4 + 2z + 5z^2 + 3z^3$

4 : "f:" $2 + 3x + 5x^2 + x^3 - x^4 + 2x^6 - 5x^7$
"g:" $1 - 2x + 5x^3$

5 : "f:" $4 + 3a - 2a^2 + 3a^3 - 5a^4 - 5a^5 + 2a^6$
"g:" $4 + 2a + 2a^2 - a^3$

6 : "f:" $-4c + c^2 + 5c^3 + 5c^4 - 5c^5 + 2c^6 - c^7$
"g:" $-4 + 2c + 2c^2$

7 : "f:" $-2a - a^2 + a^3 - 4a^4 - 2a^5 + 2a^6 + 5a^7$
"g:" $4 + 2a + 3a^2 + 4a^3$

8 : "f:" $-5 - y + 4y^2 - 3y^3 - 3y^4 - y^5 + 2y^6 - 4y^7$
"g:" $5 + 2y - 5y^2 + y^3$

9 : "f:" $5 - 2x + 5x^3 - 3x^4 + x^5 + 3x^6 - x^7$
"g:" $-5 + 3x + 4x^2 - 2x^3$

10 : "f:" $-1 + 4c + 3c^2 + 5c^3 - c^4 - 3c^5 - c^6 + 3c^7$
"g:" $-5 + c + 4c^2 + c^3$

Задание 2.2 Найдите НОД (наибольший общий делитель) и НОК (наименьшее общее кратное) многочленов f и g

а) Найдите НОД многочленов f и g, используя встроенную функцию *Mathematica* **PolynomialGCD**. Выполните предложенный вариант задания. Результат вычислений оформите в виде таблицы.

1 :	"f:"	$-5 - c - 14 c^2 + 18 c^3 + 6 c^4 + 18 c^5 + 8 c^6$	
		"g:"	$-4 - 4 c - 10 c^2 + 7 c^3 + 20 c^4 + 14 c^5 + 8 c^6 + 2 c^7$
2 :	"f:"	$12 + 29 t + 11 t^2 - 18 t^3 - 10 t^4 + 30 t^5 - 5 t^6$	
		"g:"	$8 + 10 t + 3 t^2 - 2 t^3 + 11 t^4 + 5 t^5 - 10 t^6 + 10 t^7$
3 :	"f:"	$-5 - 19 z + 18 z^2 - 11 z^3 - 8 z^4 + 32 z^5 - 15 z^6$	
		"g:"	$-5 + 11 z - 8 z^2 + 4 z^3 + 11 z^4 - 13 z^5 + 9 z^6 - 5 z^7$
4 :	"f:"	$3 + 9 b - 8 b^2 - 8 b^3 - 6 b^4 - 8 b^5 + 8 b^6$	
		"g:"	$3 b + 6 b^2 + 10 b^3 - 6 b^4 - 16 b^5 - 16 b^6 - 16 b^7$
5 :	"f:"	$4 - 11 y + y^2 - 5 y^3 + 44 y^4 - 36 y^5 + 16 y^6$	
		"g:"	$-16 - 24 y - 33 y^2 + 17 y^3 - 6 y^4 + 21 y^5 - 23 y^6 + 12 y^7$
6 :	"f:"	$8 x - 14 x^2 + 5 x^3 - 11 x^4 + 12 x^5$	
		"g:"	$20 x^2 - 21 x^3 + 21 x^4 - 27 x^5 + 19 x^6 - 12 x^7$
7 :	"f:"	$-25 x - 45 x^2 - 30 x^3 - 5 x^4 + 4 x^5 + 2 x^6$	
		"g:"	$20 x + 25 x^2 + 20 x^3 + 21 x^4 + 9 x^5 - 3 x^6 - 2 x^7$
8 :	"f:"	$-3 t + 8 t^2 + 32 t^3 - 4 t^4 + 16 t^6$	
		"g:"	$-9 t - 27 t^2 + 7 t^3 + 22 t^4 - 48 t^5 + 28 t^6 - 8 t^7$
9 :	"f:"	$2 t + 8 t^2 - t^4 + 15 t^5 - 15 t^6$	
		"g:"	$-10 t - 18 t^2 - 8 t^3 - 21 t^4 - 16 t^5 + 10 t^7$

10 : "f:" $-10-20c+5c^2+25c^3-20c^4-5c^5+25c^6$
"g:" $-8-18c-10c^2-5c^3-28c^4-6c^5-3c^6-10c^7$

б) Найдите НОК многочленов f и g, используя встроенную функцию *Mathematica* **PolynomialLCM**. Выполните предложенный вариант задания. Результат вычислений оформите в виде таблицы.

1 : "f:" $-9a-6a^2+6a^3+4a^4$
"g:" $-9a-9a^2-9a^3+9a^4+10a^5-2a^6$

2 : "f:" $2-t+2t^2-5t^3-27t^4-15t^5$
"g:" $-4-8t-15t^2-34t^3-3t^4+13t^5-15t^6$

3 : "f:" $-9+3a+12a^2-6a^3-6a^4$
"g:" $-12+19a-a^2-21a^3+14a^4+2a^5-6a^6$

4 : "f:" $16-4a+16a^2+20a^4$
"g:" $-8-24a-8a^2+8a^3+24a^4+8a^5$

5 : "f:" $-12z-4z^2+z^3-22z^4+12z^5$
"g:" $-4z+4z^2+3z^3-26z^4-6z^5+9z^6$

6 : "f:" $-12-5y^2+13y^3+6y^4+9y^5$
"g:" $-9+6y-9y^2+16y^3-4y^4+8y^5-3y^6$

7 : "f:" $9+12t-18t^2-20t^3+9t^4+5t^5$
"g:" $-6-18t-2t^2+28t^3+12t^4-10t^5-4t^6$

8 : "f:" $-6-2z-2z^2+3z^3+z^4+z^5$
"g:" $2-6z-4z^2-5z^3+3z^4+2z^5+2z^6$

9 : "f:" $12-17y+6y^2-12y^3+8y^4$
"g:" $12+7y-8y^2-3y^3-25y^4-4y^5-12y^6$

10: "f:" $-3+12y-13y^2+11y^3+3y^4$

"g:" $9-9y+9y^2+3y^3-y^4-5y^5-y^6$

Задание 2.3 Представьте рациональную функцию в виде суммы рациональных дробей

Представьте рациональную функцию в виде суммы рациональных дробей, используя встроенные функции *Mathematica* **Apart** и **ApartSquareFree**. Объясните разницу.

1. $\dfrac{1}{(-1-3x)(-5+x)(-9+x^2)^3}$
2. $\dfrac{1}{y^2(5+2y)(1+5y)(-9+y^2)^2}$
3. $\dfrac{1}{(5-4x)(-5+x)(-25+x^2)(-16+x^2)^2}$
4. $\dfrac{1}{(5-3t)(2+5t)(-16+t^2)^2(-9+t^2)}$
5. $\dfrac{1}{2(-3-z)(-9+z^2)(-1+z^2)^2}$
6. $-\dfrac{1}{5(-5-2c)c(-25+c^2)(-9+c^2)^2}$
7. $-\dfrac{1}{z(-1+4z)(-25+z^2)(-16+z^2)^2}$
8. $\dfrac{1}{(5-3b)(2+4b)(-4+b^2)(-1+b^2)^2}$
9. $\dfrac{1}{(1+a)(5+4a)(-16+a^2)^2(-4+a^2)}$
10. $\dfrac{1}{2(-5-2a)(-4+a^2)(-1+a^2)^2}$

Задание 2.4 Приведите сумму дробно-рациональных функций к общему знаменателю

Для выполнения задания используйте встроенную функцию *Mathematica* **Together**, в качестве исходной суммы -- результаты выполнения задания 2.3.

Задание 2.5 Сократите общие множители числителя и знаменателя дробно-рациональной функции

Для выполнения задания используйте встроенную функцию *Mathematica* **Cancel**.

1. $\dfrac{4-18c+34c^2-38c^3+23c^4-9c^5-10c^6+4c^7+6c^8+c^9}{-10+20c-20c^2-5c^3}$
2. $\dfrac{3-7y-18y^2-3y^3-38y^4+6y^5-8y^6+y^7+13y^8+4y^9+5y^{10}}{-3+10y+7y^2+6y^3+5y^4-2y^5+9y^6+21y^7+23y^8+26y^9+2y^{10}+15y^{11}}$
3. $\dfrac{20-a-6a^2+11a^3+14a^4-8a^5+12a^6+17a^7-16a^8+17a^9+12a^{10}}{12+17a-8a^2+7a^3+20a^4}$
4. $\dfrac{-8+20a+4a^2+12a^3}{4a^2-6a^3-4a^4-24a^5-10a^6-39a^7-11a^8-15a^9}$
5. $\dfrac{6+21c-10c^3-25c^4+13c^5-22c^6+22c^7-2c^8-c^{10}-10c^{11}+10c^{12}}{6+3c-3c^2-10c^3+5c^4}$
6. $\dfrac{-25+35a-35a^2+25a^3-9a^4+5a^5-2a^6}{-5+15a^2-18a^3+17a^4-6a^5}$
7. $\dfrac{-2+9z-22z^2+15z^3-4z^4-18z^5-30z^6-25z^7}{4-12z+16z^2+20z^3}$

8. $$\frac{-25-15 y+4 y^2-20 y^3+14 y^4+9 y^5+16 y^6+31 y^7+27 y^8-y^9+20 y^{10}}{-25-45 y+17 y^3+18 y^4+3 y^5-19 y^6+7 y^7-26 y^8-y^9+y^{10}+3 y^{11}+3 y^{12}+4 y^{13}}$$

9. $$\frac{-16+29 a^2+49 a^3-15 a^4-28 a^5+23 a^6-13 a^7-20 a^8-18 a^9-6 a^{10}+9 a^{11}}{-12+3 a+14 a^2+26 a^3-a^4-13 a^5+55 a^6-34 a^7-12 a^8+40 a^9-15 a^{10}}$$

10. $$\frac{5-24 t+18 t^3-11 t^4-7 t^5+24 t^6+16 t^7}{10+27 t-5 t^2-33 t^3-20 t^4}$$